
Contexte de l’étude
Nous évoluons aujourd’hui dans une période en pleine transition énergétique, des réflexions sur les
problèmes urbanistiques de nos villes actuelles apparaissent. La chaleur dans nos habitations n'est
pas encore optimisée et dans les zones urbaines, des îlots de chaleur se font de plus en plus
ressentir.

Des températures élevées causent de l’inconfort dans nos habitations et dépendent de plusieurs
paramètres. Nous nous concentrerons sur l’impact que peut avoir le rayonnement du soleil sur nos
parois de bâtiment.

Problématique
Notre problématique portera principalement sur les effets du rayonnement du soleil sur les façades
exposées pouvant causer une augmentation des températures en périphérie de murs et impactant
alors les températures intérieures.

Après avoir modélisé et exploité l’évolution des températures en régime établi, nous avons observé
les limites de l’analyse de cette étude. Pour pouvoir observer sur une journée l’évolution de la
température, il est nécessaire de passer notre modélisation en régime transitoire.

Généralités
Localisation

La paroi que nous étudions provient d’un bâtiment situé à
Anglet. Notre simulation de l’évolution de la température durera
tout au long de la journée la plus longue de l’année le
21/06/2020.

Nous avons par conséquent recueilli les paramètres d'élévation
et d'azimut du soleil à partir du site Sunearth Tools.

Modélisation du problème
Schéma de principe initial

Nous décidons d’étudier le
problème en 2D en
modélisant la paroi d’une
façade d’un immeuble
représentée par un mètre
linéaire de cette dernière
sans ouverture. Nous
souhaitons modéliser le flux
traversant dans la paroi et
donc étudier la température
de diffusion dans la paroi en
prenant en compte les
conditions climatiques
extérieures. Nous agissons
dans un premier temps en
régime permanent.

Géométrie

Matériaux Longueur Epaisseur

Béton 1 m 20 cm

Isolant : laine de bois 1 m 10 cm

Plaque de plâtre BA13 1 m 3 cm

Caractéristiques des matériaux

Pour notre modélisation nous avons récupéré les propriétés thermiques de chaque matériau que
nous intégrerons par la suite dans notre programme. La paroi est donc composée de béton, d’un
isolant laine de bois et d’une plaque de plâtre type BA13.

Matériaux Conductivité
thermique  (W/m.K)

Chaleur spécifique
(J /(/K.kg))

Masse volumique
(kg/m3)

Béton 0.56 385 2400

Isolant : laine de bois 0.038 2000 50

Plaque de plâtre BA13 0.25 1008 325

Phénomènes à modéliser

Ainsi nous modélisons les différents phénomènes thermiques s’appliquant sur la paroi :
- Conduction : Dans le mur traversant tous les composants (brique, isolant, plâtre)
- Convection : Sur la surface extérieure et intérieure de la paroi

Difficultés rencontrées
Ce que nous obtenons en régime stationnaire, c'est-à-dire à un instant donné, n’est pas vraiment
efficace puisque nous ne pouvons pas étudier la manière dont évolue la température au cours de la
journée. Cependant le résultat que nous obtenons dans cette simulation nous sert de base pour la
simulation en régime transitoire.

Passage en régime transitoire :

Pour modéliser notre problème en régime transitoire il a fallu prendre en compte des paramètres
variant au cours du temps tels que la température et l’énergie solaire pour l’intégrer à notre procédure
d’évolution. Toutes les valeurs sont compilées dans un tableur Excel (cf. “
Calcul_E_directionnel.xlsx”).

Il est également important de prendre en compte le changement des caractéristiques thermiques du
béton varient en fonction de la température d’ambiance. Nous avons donc fixé différentes valeurs
pour la masse volumique du béton en fonction de la température.
Evolution de la masse volumique du béton :

Température Masse volumique

20°C 2400 kg/m3

50°C 2300 kg/m3

Pour la résolution du problème en régime transitoire sur CAST3M, nous utilisons la procédure
PASAPAS permettant de calculer des valeurs sur un pas donné. Pour résoudre ce problème, nous
ferons le calcul via cette procédure toutes les dix minutes afin d’obtenir assez de valeurs et donc
affiner les résultats.

Ecriture du programme sur CAST3M
Définition du maillage

On définit le maillage en 2D avec un maillage triangulaire à 6 points pour la précision du maillage
avec une densité de 0.02 afin que la procédure ne soit pas trop lourde :

OPTI DIME 2 ELEM TRI6 DENSITE 0.02;

Pour pouvoir créer les surfaces en fonction des matériaux nous définissons leur épaisseur et la
longueur de la paroi :

L_Y = 1. ; Hauteur de la paroi en mètre.
Epmat1 = 0.2; Épaisseur du béton en mètre.
Epmat2 = 0.1; Épaisseur de l’isolant  en mètre.
Epmat3 = 0.03; Épaisseur de la plaque de plâtre en mètre.

Création des points de tracé :

P0=0. 0. ; P1=0. L_Y;

En définissant la droite D1 à partir des points P0 et P1 nous pouvons créer les surfaces par
translation suivant les vecteurs VEC X, comme l’illustre le schéma suivant:

D1 = DROI P0 P1 ;
VEC1 = Epmat1 0. ;
S1 = D1 TRANSLATION VEC1 ;
S1b = S1 COUL BRIQ ;
*
D2 = S1 COTE 3;
VEC2 = Epmat2 0. ;
S2 = D2 TRANSLATION VEC2  ;
S2b = S2 COUL BEIG ;
*
D3 = S2 COTE 3 ;
VEC3 = Epmat3 0. ;
S3 = D3 TRANSLATION VEC3  ;
S3b = S3 COUL ROUG ;
*
D4 = S3 COTE 3 ;
D5 = cote 2 S1 ;
D6 = (cote 4 S2) inve ;
D7 = cote 2 S3 ;
DT = D5 et D6 et D7 ;
P2 = D5 point final ;
P3 = D6 point final ;
P4 = D7 point final ;
*
STOT = S1b ET S2b ET S3b ;
TRACER STOT ;

Résolution en régime permanent

La première partie du programme permet de résoudre le problème sans la dimension temporelle.
Elle servira de base pour le déclenchement de la résolution en régime transitoire. Nous posons donc
les conditions limites correspondantes aux conditions de départ du futur régime transitoire le
21/06/2021 à minuit.

Paramètres

T_ext = 17.4 ; Température à l'extérieur à minuit en °C.
T_int = 21. ; Température à l’intérieur à minuit en °C.
*
H_ext = 40. ; Coefficient de convection extérieure en W/°C.
H_int = 10. ; Coefficient de convection intérieure en W/°C.
*
Cte_sb = 5.673e-8 ;

Matériaux

K_bet = 0.56 ; Conductivité thermique du béton en W/(m.K)
K_isol = 0.038 ; Conductivité thermique de l’isolant en W/(m.K)
K_plat = 0.25 ; Conductivité thermique du plâtre en W/(m.K)

Définition du modèle de conduction sur les différentes surfaces

MOD1 = MODE S1 THERMIQUE CONDUCTION ; Création du premier modèle de conduction.
MAT1 = MATE MOD1 'K' K_bet ; Application du premier modèle pour béton.
MOD2 = MODE S2 THERMIQUE CONDUCTION ; Création du second modèle de conduction.
MAT2 = MATE MOD2 'K' K_isol ; Application du second modèle pour l’isolant.
MOD3 = MODE S3 THERMIQUE CONDUCTION ; Création du troisième modèle de conduction.
MAT3 = MATE MOD3 'K' K_plat; Application du troisième modèle pour l’isolant.

Définition du modèle de convection sur les différentes surfaces

MOCONV1 = MODE D1 THERMIQUE 'CONVECTION' ; Création du premier mode de convection.
MACONV1 = MATE MOCONV1 'H' H_ext ; Application sur le matériau du coef extérieur.
MOCONV2 = MODE D4 THERMIQUE 'CONVECTION' ; Création du second mode de convection.
MACONV2 = MATE MOCONV2 'H' H_int ; Application sur le matériau du coef intérieur.

Définition du modèle complet

MOTOT = MOD1 ET MOD2 ET MOD3 ET MOCONV1 ET MOCONV2 ;
Assemblage des différents modes de convection.
MATOT = MAT1 ET MAT2 ET MAT3 ET MACONV1 ET MACONV2 ;
Assemblage de la convection sur les matériaux concernés.

Définition des conditions limites

On définit les conditions limites des flux de convection à l’extérieur et à l’intérieur pour pouvoir
lancer le calcul. La résolution avec des initialisations de flux est en fonction de la température fixée
préalablement.

FL_ext = CONVECTION MOCONV1 MACONV1 'T' T_ext ;
FL_int = CONVECTION MOCONV2 MACONV2 'T' T_int ;

On définit les conditions limites des températures sur les surfaces des matériaux donnant sur
l’intérieur et l’extérieur pour avoir une condition limite de conduction.

COND_bet = BLOQ D1 'T' ;
FL_ext = DEPI COND_bet T_ext ;
COND_plat = BLOQ D4 'T' ;
FL_int = DEPI COND_plat T_int ;

Définition de la matrice de résolution

COND1 = CONDUC MOTOT MATOT ;
CONDTOT = COND1 ET COND_bet Et COND_plat ;
Cette matrice regroupe les équations de conduction de chaque matériau.

Définition du second membre

On exprime le second membre correspondant aux conditions limites de flux pour permettre la
résolution du système.

FLTOT = FL_ext ET FL_int ;

Expression de la solution
Castem résout la matrice température le long de la paroi, cette matrice sera utilisée pour la résolution
du problème en régime transitoire.

TEMP1 = RESOU CONDTOT FLTOT ;

Résolution en régime transitoire

Après avoir obtenu les résultats pour le problème permanent nous passons à la modélisation de notre
problème en régime transitoire en commençant par prendre en compte les paramètres d’évolution de
rayonnement.

On récupère la liste de la position du soleil et en connaissant le coefficient de visibilité (entre 1 et 0) et
la taux d’irradiance s’élevant à 1367 W/m² nous pouvons en déduire l’évolution de l’énergie solaire.
La démarche détaillée du calcul est détaillée dans un fichier ci-joint. (cf. “
Calcul_E_directionnel.xlsx”).

Rayonnement solaire

Nous créons en premier lieu une liste correspondant aux valeurs de l’énergie solaire calculées
précédemment dans le excel. La démarche est détaillée

Listemis = prog 0. 0. 0. 0. 0. 0. -13.15 76.20 259.52 457.12 641.23 778.64 817.41 645.22 95.57
-538.83 -799.19 -801.69 -684.05 -508.68 -311.95 -122.30 13.15 0. 0. 0.;
*
listemis2 = (listemis + (ABS listemis)) / 2. ;

Les valeurs égale à 0 dépendent du coefficient de visibilité lumineux égal à 0, durant toute la nuit. Les
valeurs négatives illustrent un phénomène d’ombre produit par le bâtiment lui-même sur la façade
exposée au soleil. “Listemis2” permet de ne pas prendre en compte ces valeurs négatives. Cette liste
est en relation avec la liste de temps qui guidera la procédure PASAPAS que nous verrons plus tard
dans le code.

Evolution des caractéristiques thermiques en fonction de la température

* Béton
K_bet = 0.56 ; Coefficient de conductivité thermique du béton supposé constant en
W/(m.K).
CP_bet = 385. ; Coefficient de chaleur spécifique du béton supposé constant en J/(kg.K).

Les masses volumiques varient en fonction de la température :
list_temp_rho = PROG 20. 50. ;
Liste des températures remarquables pour le changement de la masse volumique du béton.
list_rho_bet = PROG 2400. 2300. ;
Liste des masses volumiques du béton qui correspondent respectivement aux températures de la liste
précédente.
RHO_bet = EVOL MANU 'T' list_temp_rho 'RHO' list_rho_bet ;
Cette évolution est définie grâce à la procédure EVOL MANU.

Nous effectuons la même démarche pour l’isolant et le plâtre.
* Isolant
K_isol = 0.038 ;
CP_isol = 2000. ;
list_rho_isol = PROG 50. 40. ;
RHO_isol = EVOL MANU 'T' list_temp_rho 'RHO' list_rho_isol ;
*
* Plâtre
K_plat = 0.25 ;
CP_plat = 1008. ;
list_rho_plat = PROG 825. 750. ;
RHO_plat = EVOL MANU 'T' list_temp_rho 'RHO' list_rho_plat ;

De la même manière que pour la résolution en régime permanent, nous devons définir les
phénomènes thermiques s’appliquant à notre paroi en prenant cette fois ci les nouvelles valeurs des
masses volumiques. Cela concerne la conduction, la convection et l’assemblage du modèle entier.

Définition du modèle de conduction sur les différentes surfaces

MO1 = MODE S1 THERMIQUE CONDUCTION ;
MA1 = MATE MO1 'K' K_bet 'C' CP_bet 'RHO' Rho_bet ;
MO2 = MODE S2 THERMIQUE CONDUCTION ;
MA2 = MATE MO2 'K' K_isol 'C' CP_isol 'RHO' RHO_isol ;
MO3 = MODE S3 THERMIQUE CONDUCTION ;
MA3 = MATE MO3 'K' K_plat 'C' CP_plat 'RHO' RHO_plat ;

Définition du modèle de convection sur les différentes surfaces

MOCOV1 = MODE D1 THERMIQUE 'CONVECTION' ;

MACOV1 = MATE MOCOV1 'H' H_ext ;
MOCOV2 = MODE D4 THERMIQUE 'CONVECTION' ;
MACOV2 = MATE MOCOV2 'H' H_int ;

Définition du modèle complet

MOTO = MO1 ET MO2 ET MO3 ET MOCOV1 ET MOCOV2  ;
MATO = MA1 ET MA2 ET MA3 ET MACOV1 ET MACOV2 ;

Initialisation de la procédure PASAPAS

Après avoir défini les conditions de convection et de conduction sur toute la paroi en fonction des
éléments nous pouvons initialiser la procédure PASAPAS.

Définition de l’évolution de la température en fonction du moment dans la journée

L’étude se déroule le 21 juin de minuit à 23h sur chaque heure de la journée. Les heures de lever et
de coucher de soleil sont ajoutées. La conversion s’effectue en secondes, soit l’unité scientifique
utilisée par le logiciel (3600 secondes = 1 h).
Listtemps = prog 0. 3600. 7200. 10800. 14400. 18000. 19517. 21600. 25200. 28800. 32400. 36000.
39600. 43200. 46800. 50400. 54000. 57600. 61200. 64800. 68400. 72000. 75041. 75600. 79200.
82800. ;
Liste des températures correspondant à chaque heure précédemment sélectionnées.
listtext = prog 17.4 16.5 15.9 15.2 14.5 14.3 14.2 14.1 13. 15.6 18.4 21. 22.5 23.6 24.1 23.8 23.2
22.5 22.2 22. 21.4 20.2 20.2 20.2 19.2 18.6 ;

Calcul de l’évolution de la température extérieure et du rayonnement émis au cours de la journée
grâce aux listes définies précédemment.

evoltext = evol manu 't' listtemps 'T' listtext ;
evolemis = evol manu 't' listtemps 'E' listemis2 ;

Définition des flux s'appliquant sur la paroi

Le flux provenant de l’extérieur se diffuse naturellement sur la paroi en fonction de la température
extérieure :

FLext = MANU CHPO D1 1 'T' 1 nature diffus ; Définition du flux et de sa nature

cht1 = char 'TECO' evolText FLext ; Définition du chargement correspondant au flux extérieur.

Le flux l’intérieur se diffuse sur la paroi en fonction de la température intérieure : cette dernière étant
constante nous faisons donc évoluer le temps. Cette évolution va de 0 à 82 800 s (représentant les
heures) en imposant une température Tint de manière constante tout au long de la journée.

progt = (prog 0. pas 3600. 82800.) ; Démarre à 0s avec un pas de 3600s jusqu’à 82800s.
n1 = dime progt; Dimension de la liste progt.
prog1 = prog 5. pas 0.5 30.; Démarre à 5°C, avec un pas de 0,5°C jusqu’à 30°C.
tfin = extr progt n1;
prog1 = prog 0. tfin; Démarre de 0 à la valeur tfin.
prog2 = prog T_int T_int; Démarre de T_int à la valeur T_int (car constante).

EVOL2 = EVOL MANU PROG1 PROG2; Évolution des listes grâce à cette procédure.
FLint = MANU CHPO D4 1 'T' 1 nature diffus ; Flux s’appliquant à D4 (interface intérieure)
cht2 = char 'TECO' EVOL2 FLint ; Création du chargement du flux intérieur.

Le flux rayonnant s’appliquant sur la droite D1 c’est à dire sur la façade de la paroi exposée au soleil :
FL_ray = FLUX MO1 1 D1 ;
cht3 = char 'Q' evolemis FL_ray ; Création du chargement du rayonnement émis.

Initialisation des tables

Nous créons des tables correspondant aux différentes caractéristiques du problème à savoir :
❖ le modèle;
❖ les caractéristiques des matériaux;
❖ le chargement correspondant aux flux appliqués;
❖ la période sur laquelle nous voulons faire l’expérience.

tabnl = table ;
tabnl . 'MODELE' = MOTO ;
tabnl . 'CARACTERISTIQUES' = MATO ;
tabnl . 'CHARGEMENT' = cht1 et cht2 et cht3;
tabnl . 'TEMPS_CALCULES' = (prog 0. pas 600. 82800.) ; Le test dure 82800s avec un pas toutes les
10 min (600s). Ce pas permet d’avoir une meilleure précision sur les résultats finaux.
tabnl . 'TEMPERATURES' = table ;
tabnl . TEMPERATURES.0 = TEMP1 ; Initialisation de la température avec le résultat trouvé en
régime permanent.
tabnl . 'PROCEDURE_THERMIQUE' = 'DUPONT' ; Application de la procédure thermique
permettant d’obtenir les résultats sous forme de température.

Déclenchement de la procédure

pasapas tabnl ; Appel de la procédure PASAPAS

Création de supports pour le post traitement

Evolution de la T dans le mur à différents points du maillage :
La procédure évolution permet de produire les calculs à chaque point au cours du temps en créant
une courbe pour chaque point (P1,P2,P3,P4) correspondant aux interfaces des éléments dans la
paroi.

EVO1 = EVOL BLEU TEMP tabnl . TEMPERATURES 'T' P1 ;
EVO2 = EVOL GRIS TEMP tabnl . TEMPERATURES 'T' P2 ;
EVO3 = EVOL BEIG TEMP tabnl . TEMPERATURES 'T' P3 ;
EVO4 = EVOL ROUG TEMP tabnl . TEMPERATURES 'T' P4 ;

On crée une table pour ajouter les légendes et le marquages des courbes.

TAB1     = 'TABLE' ;
TAB1 . 1 = 'MARQ SS CROI ';
TAB1 . 2 = ' TIRR ';
TAB1 . 3 = ' TIRC';
TAB1 . 4 = ' TIRM';

TAB1 . 'TITRE'     = 'TABLE' ;

TAB1 . 'TITRE'. 1     =  'Interface Ext Beton' ;
TAB1 . 'TITRE'. 2     =  'Interface Beton Isolant' ;
TAB1 . 'TITRE'. 3     =  'Interface Isolant Platre' ;
TAB1 . 'TITRE'. 4     =  'Interface Platre Int' ;
'DESS' (EVO1 'ET' EVO2 'ET' EVO3 'ET' EVO4) 'LOGO' 'LEGE' TAB1
'TITR' 'Evolution de la temperature a l interface des differents materiaux' 'TITX' 't(s)' 'POSX' 'CENT'
'TITY' 'T (°C)' 'POSY' 'CENT' 'GRIL' 'POIN' 'GRIS';

Evolution de la T dans le mur à différentes heures de la journée :
Cette fois ci nous décidons de créer un drive à différents moment de la journée, nous créons donc
autant de procédure EVOL que nous voulons de courbes correspondant à un moment de la journée.
*
EVOL1 = EVOL BLEU CHPO tabnl.temperatures.1 'T' DT ;
EVOL2 = EVOL VERT CHPO tabnl.temperatures.34 'T' DT ;
EVOL3 = EVOL ROUG CHPO tabnl.temperatures.43 'T' DT ;
EVOL4 = EVOL GRIS CHPO tabnl.temperatures.67 'T' DT ;
EVOL5 = EVOL INDI CHPO tabnl.temperatures.79 'T' DT ;
EVOL6 = EVOL ORAN CHPO tabnl.temperatures.97 'T' DT ;
EVOL7 = EVOL CHPO tabnl.temperatures.126 'T' DT ;
*
TAB2     = 'TABLE' ;
TAB2 . 'TITRE'     = 'TABLE' ;
TAB2 . 'TITRE'. 1     =  '00h Initialisation' ;
TAB2 . 'TITRE'. 2     =  '5h30 lever du soleil' ;
TAB2 . 'TITRE'. 3     =  '7h Text la plus basse' ;
TAB2 . 'TITRE'. 4     =  '11h Emissivite la plus importante' ;
TAB2 . 'TITRE'. 5     =  '13h Text la plus haute' ;
TAB2 . 'TITRE'. 6    =  '16h Moment le plus ombrage' ;
TAB2 . 'TITRE'. 7    =  '20h50 Coucher du soleil' ;
'DESS' (EVOL1 ET EVOL2 ET EVOL3 ET EVOL4 ET EVOl5 ET EVOL6 ET EVOL7) 'TITR' 'Evolution
de la temperature dans le mur a differentes heures de la journee' 'LOGO' 'LEGE' TAB2 'TITX' 'ep(m)'
'POSX' 'CENT' 'TITY' 'T(°C)' 'POSY' 'CENT' 'GRIL' 'POIN' 'GRIS';

Création du fichier .ps pour la création d’un GIF :
On trace à chaque heure de la journée l’évolution de la température dans la paroi. Chaque tracé
s’affiche sous un tracé ‘.ps’ qui s’enregistre dans le dossier ‘Projet.ps’.

OPTI TRAC PSC FTRA 'Projet.ps' ;
TRAC tabnl.TEMPERATURES.1 STOT 'TITR' 'Evolution de la temperature dans le mur a 00h';

.

.

.
TRAC tabnl.TEMPERATURES.138 STOT 'TITR' 'Evolution de la temperature dans le mur a 23h'
Dans la commande de l’ordinateur, nous appelons la commande Convert du logiciel GIMP. Les
images s'enregistrent à l’envers dans le dossier des TRAC extraits de Cast3m. On demande alors
d’abord de retourner l’image puis de convertir le dossier .ps en .gif permettant d’obtenir le GIF
ci-dessous.
* dans cmd / convert -rotate 90 -dispose Background Projet.ps Projet.gif

Post-traitement
Evolution de la température dans la paroi à différents moments

de la journée du 21/06/2020

Pour analyser nos résultats, nous sélectionnons les températures qui nous semblent les plus
significatives. Dans l’ordre d’arrivée, nous étudions l’évolution des températures à minuit pour
l’initialisation, l’opposition entre le lever et le coucher de soleil, celle du moment de la journée où la
température extérieure est la plus élevée et la plus basse ainsi que celle de l’instant où l'émissivité
sur le mur est la plus grande face et celui où il est le plus ombragé.

Courbe bleue : À minuit la T° extérieure est à 17,4 °C, elle est donc inférieure à la température
intérieure T°int = 21°C; dans ce cas là, le flux de chaleur va du chaud vers le froid vers le froid. La
variation de température la plus significative dans le mur est au niveau de l’isolant, qui assure son
rôle de barrière thermique peu conductrice. Au niveau du plâtre la variation est minime, il n’a pas
d’effet résistant thermique sur le mur.

Courbe verte : À 5h30 à une T° extérieure de 14,2 °C, le soleil se lève. Cette courbe est très
semblable à la courbe bleue ayant presque les mêmes paramètres. Il s’agit de la courbe de
température la plus inférieure de l’étude.

Courbe rouge : À 7h du matin avec la T° extérieure la plus basse à 13°C. Cette courbe à l’interface
du béton à une température supérieure aux courbes bleue et verte. Cela s’explique par l’effet du
rayonnement solaire s’appliquant sur la façade. Au niveau de l’isolant et du plâtre les courbes sont
similaires aux précédentes.

Courbe grise : À 11h, à une T° de 18,4 °C lorsque l’émissivité solaire est la plus importante. Cette
courbe de T° est largement supérieure aux autres atteignant 37°C. Elle illustre l’impact le plus flagrant
du rayonnement solaire dans cette étude.

Courbe violette : À 13h, avec la T° extérieure la plus élevée à 24,1 °C, on obtient la deuxième
courbe en termes de T° élevée au niveau de la paroi extérieure du béton. Avec 6°C de différence
entre la courbe grise et la courbe jaune, on peut déduire que la T° extérieure n’est pas le facteur qui à
l’influence la plus marquante mais bien l’émissivité de la paroi qui dépend du rayonnement solaire.

Courbe orange : À 16h, avec une T° extérieure de 22,5°C, on se situe à l’instant où la paroi est la
plus ombragée. La différence entre la T° extérieure et intérieure n’étant que de 1,5 °C, l'échange du
flux n’est pas très important et nous observons également que la T° maximale se trouve dans le
béton près de son interface entre l’isolant. L’inertie thermique du béton est illustrée par cette hausse
de T°, le matériau conserve la chaleur dans sa masse et la relâchera pendant lorsque les T°
baisseront.

Courbe noire : À 20h50, à 20,2°C, lors du coucher de soleil, est obtenue une allure de courbe
semblable à la courbe orange. Nous déduisons donc que la présence ou l'absence du soleil n’impacte
pas directement la T° extérieure. Cela se fait ressentir progressivement avec l’augmentation et la
baisse des températures tout au long de la course du soleil.

Evolution de la température aux interfaces des différents

matériaux pendant toute la journée

En se plaçant au niveau des interfaces des différents matériaux constituant la paroi, nous pouvons
étudier la réaction des éléments entre eux et leur influence sur la température de l’interface intérieure.

A l’interface extérieur - béton (courbe bleue) : Nous observons la manière dont évolue la
température extérieure dépendante du rayonnement solaire. Il apporte une énergie non négligeable
directement sur la paroi. C’est donc en milieu de journée que la température est la plus élevée
atteignant  les 37°C en surface de paroi.

A l’interface béton - isolant (courbe grise) : Cette représentation montre bien l’effet qu’a l’isolant
sur la paroi. Avec l’évolution de la différence entre T° intérieure imposée à 21°C et la T° extérieure, le
flux de chaleur ne va pas dans le même sens. Les valeurs extrêmes de la courbe de T° à l’intérieur
de l’isolant illustre le changement de direction du flux. Le rôle de l’isolant est de stopper le
mouvement de fraîcheur intérieure et l’arrivée de la chaleur extérieure.
Par exemple à 40000 secondes soit à 11h, lorsque la température est la plus élevée, l’isolant retient
la chaleur provenant du béton  (en référence au pic de la courbe bleue).

A l’interface isolant - plâtre (courbe beige) et plâtre - intérieur (courbe rouge) : Nous
remarquons que l’évolution à l’interface entre le plâtre et l’isolant est semblable à l’interface plâtre et
l’intérieur, nous pouvons donc en déduire que le plâtre n’a aucun avantage thermique.

Création d’un GIF pour illustrer l’évolution de la température dans

la paroi durant une journée

Comme nous souhaitions exploiter l’évolution de température dans le mur au cours de la journée, il
nous a semblé plus visuel de la présenter sous forme de GIF.

Nous remarquons que de 0h à 7h, le flux de chaleur va de l’intérieur vers l’extérieur puis à partir de 8h
change de sens car la T° extérieure est supérieure à celle de l'intérieur pour frôler les 40°C. La T° la
plus élevée se trouve généralement aux interfaces intérieures et extérieures sauf de 16h à 22h où elle
se trouve au milieu du mur. Cela s’explique par la différence entre les T° extérieures et intérieures très
faibles ainsi que la chaleur se trouvant principalement stockée à l'intérieur du matériau béton grâce à
son inertie.

Etude de l’impact sur le changement de la densité

En augmentant la densité de notre maillage nous pouvons obtenir une coupe réaliste qui distingue
bien les couches avec les différentes couleurs. Cependant, il faut un ordinateur assez puissant pour
pouvoir réaliser le calcul avec une densité importante. En effet, l'augmentation de la densité induit une
augmentation des inconnues et des équations et rend donc le calcul encore plus long.

